



Calcul d'aires planes

© Pierre Lantagne

Enseignant retraité du Collège de Maisonneuve

Ce document est une révision de celui produit en 2005. L'objectif principal de ce document Maple est de rendre l'élève capable d'appliquer le théorème fondamental du calcul pour calculer l'aire de régions planes. À l'aide de certaines macro-commandes de la bibliothèque maison *riemann*, l'élève pourra colorier la région plane dont il faut calculer l'aire ainsi que superposer, dans un même graphique, des rectangles représentatifs des éléments différentiels d'aire. Dans ce document Maple, plusieurs exemples de calculs d'aires selon un découpage horizontal et/ou vertical sont développés afin de proposer une bonne manière de documenter des développements Maple.

Bonne lecture à tous !

* Ce document Maple est exécutable avec la version 2020.1

Initialisation

```
> restart;  
> with(plots,display,setoptions);  
  setoptions(labels=[x,y],tickmarks=[12,12],  
            axesfont=[TIMES,ROMAN,8],labelfont=[TIMES,ROMAN,8],  
            legendstyle=[font=["TIMES", 8]],  
            size=[300,300]);  
                                [display,setoptions]
```

(1)

Macro-commandes de l'extension riemann

Initialisation

La plupart des macro-commandes utilisées dans les exemples font parties de la bibliothèque *riemann* que j'ai écrits pour mes étudiants. Pour la séance Maple qui est en cours, vous devrez donc indiquer à Maple où aller lire les instructions composant les macro-commandes de cette bibliothèque. Vous devrez donc taper dans une zone de requêtes la requête suivante:

```
with(riemann);
```

Mais, préalablement, vous devrez indiquer à Maple le chemin à suivre pour accéder aux macro-commandes de cette bibliothèque, c'est-à-dire accéder au fichier contenant le code de ces macro-commandes. Il faut donc informer Maple du chemin menant au dossier contenant le fichier dans lequel il trouvera cette bibliothèque. Pour disposer des macro-commandes sur votre ordinateur, il vous faudra procéder à l'installation de deux fichiers « riemann.ind » et « riemann.lib » sur votre disque rigide: celui de votre ordinateur personnel et non pas celui du poste sur lequel vous travaillez au collège.

À partir de mon site internet,, cliquer sur l'onglet «NYB», localiser et télécharger les fichiers « riemann.ind » et « riemann.lib » et les déposer dans un sous-dossier de votre ordinateur. Je vous suggère de nommer ce sous-dossier « Bibliothèque Riemann » et un bon choix d'emplacement pour ce dossier serait de le créer dans le dossier *Users* de votre installation Maple.

Ensuite, la façon d'informer Maple du chemin dont il est question consiste à **ajouter** à la variable d'environnement **libname** ce chemin. Mais d'abord, l'énoncé du chemin doit être une **chaîne de caractères** de la forme

```
"Nom_du_volume:\\Répertoire_1\\Répertoire_2\\...../Répertoire_contenant_le_fichier"
```

ATTENTION: La variable **libname** possède déjà une assignation qui a été faite au moment de l'installation du logiciel, soit l'assignation du chemin menant à la bibliothèque de base **maple.lib** contenu dans le dossier lib. Voyez:

```
> libname;  
"C:\Program Files\Maple 2020\lib" (2)
```

Il ne faut surtout pas écraser le contenu de la variable **libname** par une nouvelle assignation, Maple ne saurait plus comment localiser les macro-commandes de la bibliothèque de base maple.lib. La façon dont il faut s'y prendre est la suivante:

```
libname := nouveau_chemin, libname;
```

Par exemple, l'assignation suivante est celle que je dois faire sur mon ordinateur personnel.

```
> libname:="C:\\Users\\plant\\Desktop\\Éléments site Maple au  
cégep\\Downloads\\Documents Maple\\NYB/Bibliothèque riemann",  
libname;  
libname := (3)  
"C:\Users\plant\Desktop\Éléments site Maple au cégep\Downloads\Documents  
Maple\NYB/Bibliothèque riemann", "C:\Program Files\Maple 2020\lib"
```

On remarquera le dernier élément dans le résultat de cette assignation: c'est le chemin menant à la bibliothèque de base, c'est-à-dire que c'est le contenu qu'avait la variable *libname*. Cette façon de faire pour ajouter (concaténer) un nouveau terme à une séquence est valable quelque soit la variable en cause. Dans ce cas-ci, la variable en cause est la variable *libname*.

Maintenant, nous pouvons indiquer à Maple la manière d'accéder aux instructions des 18 macro-commandes de la bibliothèque *riemann*.

```
> with(riemann); (4)  
[coloriage, plot2ch, plot2cv, plot2dx, plot2dy, plot3dx, plot3dy, plotch, plotcoq, plotcoqanim,  
plotcv, plotrecth, plotrectv, plotriemann, plotrotx, plotrotxanim, plotroty, plotrotyanim]
```

plotcv


La macro-commande

```
plotcv([x, f(x), x=a..b], x=c..d, options)
```

possède deux arguments obligatoires permettant de tracer la courbe d'équation $y = f(x)$ en couleur navy sur un intervalle $[a, b]$ et de colorer la région comprise entre la courbe et l'axe des x sur un intervalle $[c, d]$. L'intervalle $[c, d]$ étant bien sûr un sous-ensemble de l'intervalle $[a, b]$. La couleur turquoise sera utilisée pour la région au-dessus de l'axe des x et la couleur rose pour celle au-dessous de l'axe des x .

L'utilisateur peut choisir un nom différent du nom x pour désigner l'abscisse.

```
> plotcv([t,t^2-4,t=-3..4],t=-2..3);
```



plotch


La macro-commande

```
plotch([f(y),y,y=a..b],y=c..d,options)
```

possède deux arguments obligatoires permettant de tracer la courbe d'équation $x = f(y)$ en couleur navy sur un intervalle $[a,b]$ de l'axe des y et de colorer la région comprise entre la courbe et l'axe des y sur un intervalle $[c,d]$. L'intervalle $[c,d]$ étant bien sûr un sous-ensemble de l'intervalle $[a,b]$. La couleur turquoise sera utilisée pour la région à la droite de l'axe des y et la couleur rose pour celle à la gauche de l'axe des y .

L'expression $f(y)$ spécifiant l'abscisse des points du graphe f doit être une expression formulée en termes de la variable y .

```
> plotch([y^2-4,y,y=-3..4],y=-3..4);
```



plot2cv

La macro-commande


```
plot2cv([f(x),g(x),x=a..b],x=c..d,options)
```

possède deux arguments obligatoires permettant de superposer les tracés des courbes d'équation $y = f(x)$ et $y = g(x)$ sur un intervalle $[a,b]$ de l'axe des x et de colorer la région comprise entre les deux courbes sur un intervalle $[c,d]$. L'intervalle $[c,d]$ étant bien sûr un sous-ensemble de l'intervalle $[a,b]$. La couleur cyan sera utilisée pour colorer la région comprise entre les deux courbes.


Les expressions $f(x)$ et $g(x)$ spécifiant l'ordonnée des points des graphes f et g doivent être des

expressions formulées en termes de la variable x .

```
> f:=x->12-3*x^2:  
g:=x->-x^2+4:  
plot2cv([f(x),g(x)],x=-2.5..2.5,x=-2..2);
```



```
> f:=x->x^2-4:  
g:=x->0:  
plot2cv([f(x),g(x)],x=-4..4,x=-4..4);
```



plot2ch

La macro-commande

```
plot2ch([f(y),g(y)],x=a..b,options)
```

possède deux arguments obligatoires permettant de superposer les tracés des courbes d'équation $x = f(y)$ et $x = g(y)$ sur un intervalle $[a,b]$ de l'axe des y et de colorer la région comprise entre les deux courbes sur un intervalle $[c,d]$. L'intervalle $[c,d]$ étant bien sûr un sous-ensemble de l'intervalle $[a,b]$. La couleur cyan sera utilisée pour colorer la région comprise entre les deux courbes.


Les expressions $f(y)$ et $g(y)$ spécifiant l'abscisse des points des graphes f et g doivent être des expressions formulées en termes de la variable y .

```
> f:=y->y^2-16:  
g:=y->16-y:  
Sol:=solve(f(y)=g(y)); # pour obtenir en y les limites du  
coloriage
```


(5)

$$\text{Sol} := -\frac{1}{2} + \frac{\sqrt{129}}{2}, -\frac{1}{2} - \frac{\sqrt{129}}{2}$$

```
> plot2ch([f(y),g(y),y=-7..7],y=min(Sol)..max(Sol));
```



```
> f:=y->sqrt(y*(5-y)^2):
g:=y->-sqrt(y*(5-y)^2):
plot2ch([f(y),g(y),y=0..6],y=0..5);
```



plotrectv

La macro-commande


plotrectv([f(x),g(x)],x=c,delta=d,options)

possède trois arguments obligatoires permettant de tracer verticalement un rectangle représentatif de l'élément différentiel d'aire entre deux courbes d'équation $y = f(x)$ et $y = g(x)$. L'équation $x = c$ spécifie l'axe de symétrie perpendiculaire à l'axe des x . Le rectangle tracé possèdera une largeur égale à $0,2d$ unité(s) et sera de couleur rouge.


Les expressions $f(x)$ et $g(x)$ spécifiant l'ordonnée des points des graphes f et g doivent être des expressions formulées en termes de la variable x .

```
> f:=x->sqrt(abs(x)):
g:=x->x+1:
Région:=plot2cv([f(x),g(x),x=-4..4],x=-4..4):Rectangle1:=plotrectv
([f(x),g(x)],x=-2,delta=3/4):
Rectangle2:=plotrectv([f(x),g(x)],x=-1/8,delta=3/8):
```

```
Rectangle3:=plotrectv([f(x),g(x)],x=3,delta=3/4):
display(Rectangle1,Rectangle2,Rectangle3,Région);
```



```
> f:=y->sqrt(abs(y)):
g:=y->2:
Région:=plot2ch([f(y),g(y),y=-4..4],y=-4..4):
Région;
```




Afin d'utiliser la macro-commande *plotrectv*, nous devons absolument expliciter l'expression $f(y)$ en terme de la variable x .

```
> isolate(f(y)=x,y);
```

$$y = \begin{cases} [0] & x = 0 \\ [-x^2, x^2] & \text{otherwise} \end{cases}$$

(6)

```
> Rectangle:=plotrectv([-x^2,x^2],x=3/2,delta=1/4):
display(Rectangle,Région);
```



plotrecth

La macro-commande


plotrecth([f(y),g(y)],y=c,delta=d,options)

possède trois arguments obligatoires permettant de tracer horizontalement un rectangle représentatif de l'élément différentiel d'aire entre deux courbes d'équation $x = f(y)$ et $x = g(y)$. L'équation $y = c$ spécifie l'axe de symétrie perpendiculaire à l'axe des y . Le rectangle tracé possèdera une largeur égale à $0,2d$ unité(s) et sera de couleur rouge.


Les expressions $f(y)$ et $g(y)$ spécifiant l'abscisse des points des graphes f et g doivent être des expressions formulées en termes de la variable y .

```
> f:=y->y^2-16:  
g:=y->16-y:  
Sol:=solve(f(y)=g(y)); # pour trouver en y les limites du  
coloriage  
Région:=plot2ch([f(y),g(y)],y=-7..7,y=min(Sol)..max(Sol)):  
Région;
```

$$Sol := -\frac{1}{2} + \frac{\sqrt{129}}{2}, -\frac{1}{2} - \frac{\sqrt{129}}{2}$$



```
> Rectangle:=plotrecth([f(y),g(y)],y=-2.5,delta=1.5):  
display(Rectangle,Région);
```



Afin d'utiliser la macro-commande **plotrectv**, nous devons absolument expliciter les expressions $f(y)$ et g

(y) en terme de la variable x.

```
> solve(f(y)=x,y);
```

$$\sqrt{x+16}, -\sqrt{x+16}$$

(7)


Aux deux points d'abscisse $x = -5$ par exemple, il est graphiquement clair que l'ordonnée y est tel que $-\sqrt{16+x} < y < \sqrt{16+x}$.

```
> Rectangle1:=plotrectv([-sqrt(16+x),sqrt(16+x)],x=-5,delta=4):
```

Aux deux points d'abscisse $x = 15$ par exemple, il est tout aussi graphiquement clair que l'ordonnée y est tel que $-\sqrt{16+x} < y < 16-x$.

```
> Rectangle2:=plotrectv([-sqrt(16+x),16-x],x=15,delta=4):
```

```
> display(Rectangle1,Rectangle2,Région);
```



Calcul d'aire d'une région plane selon un découpage vertical

Exemple 1

Aire d'une région entre la courbe et l'axe des x d'une fonction non négative sur un intervalle donné $[a,b]$


Calcul de l'aire de la région comprise entre la courbe $y = x^2$, l'axe des x et les droites $x = 1$ et $x = 4$.

Traçons d'abord la région dont nous avons à calculer l'aire.

```
> f:=x->x^2:
```


```
Région:=plotcv([t,f(t),t=-2..4.5],t=1..4):
```

```
Région;
```



Superposons le tracé d'un rectangle représentatif de l'élément différentiel d'aire.

```
> Rectangle:=plotrectv([0,f(x)],x=3,delta=2/3):
display([Rectangle,Région],view=[-2..4.5,0..20]);
```



Calculons l'aire demandée.

```
> Intégrande:=f(x)-0:
Aire:=Int(Intégrande,x=1..4)*(u^2);
```

$$Aire := \left(\int_1^4 x^2 dx \right) u^2 \quad (8)$$

```
> ``=value(Aire);
```

$$= 21 u^2 \quad (9)$$


Exemple 2

Aire d'une région entre la courbe et l'axe des x d'une fonction sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre la courbe d'équation $y = 5 + x - x^2$, l'axe des x et entre les droites d'équation $x = -4$ et $x = 4$.


Traçons d'abord la région dont nous avons à calculer l'aire.

```
> f:=x->5+x-x^2:
Région:=plotcv([x,f(x)],x=-5..5],x=-4..4):
Région;
```



Nous superposerons les tracés de trois rectangles représentatifs d'élément différentiel d'aire car nous devons tenir compte du sous-intervalle sur lequel ils seront placés.

```
> Rectangle1:=plotrectv([0,f(x)],x=-3,delta=.5):
  Rectangle2:=plotrectv([0,f(x)],x=1.75,delta=.5):
  Rectangle3:=plotrectv([0,f(x)],x=7/2,delta=.5):
  display([Rectangle1,Rectangle2,Rectangle3,Région],view=[-4..5,min
    (f(-4),f(5))..f(1/2)]);
```



Pour poser correctement les bornes de chaque intégrale définie, nous devons obtenir l'abscisse des points d'intersection de la parabole avec l'axe des x.

```
> Sol:=solve(f(x)=0,x);
```

$$Sol := \frac{1}{2} - \frac{\sqrt{21}}{2}, \frac{1}{2} + \frac{\sqrt{21}}{2} \quad (10)$$

Posons maintenant l'aire cherchée en additionnant l'aire de la surface de trois régions.

```
> Intégrandel:=0-f(x):
  Aire1:=Int(Intégrandel,x=-3..min(Sol))*(u^2);
```

$$Aire1 := \left(\int_{-3}^{\frac{1}{2} - \frac{\sqrt{21}}{2}} (x^2 - x - 5) dx \right) u^2 \quad (11)$$

```
> Intégrande2:=f(x)-0:
  Aire2:=Int(Intégrande2,x=min(Sol)..max(Sol))*(u^2);
```

(12)

$$\text{Aire2} := \left(\int_{\frac{1}{2} - \frac{\sqrt{21}}{2}}^{\frac{1}{2} + \frac{\sqrt{21}}{2}} (-x^2 + x + 5) dx \right) u^2 \quad (12)$$

> **Intégrande3:=0-f(x):**

Aire3:=Int(Intégrande3,x=max(Sol)..4)*(u^2);

$$\text{Aire3} := \left(\int_{\frac{1}{2} + \frac{\sqrt{21}}{2}}^4 (x^2 - x - 5) dx \right) u^2 \quad (13)$$

Calculons la somme suivante.

> **Aire_cherchée:=Aire1 + Aire2 + Aire3;**

$$\text{Aire_cherchée} := \left(\int_{-3}^{\frac{1}{2} - \frac{\sqrt{21}}{2}} (x^2 - x - 5) dx \right) u^2 + \left(\int_{\frac{1}{2} - \frac{\sqrt{21}}{2}}^{\frac{1}{2} + \frac{\sqrt{21}}{2}} (-x^2 + x + 5) dx \right) u^2 + \left(\int_{\frac{1}{2} + \frac{\sqrt{21}}{2}}^4 (x^2 - x - 5) dx \right) u^2 \quad (14)$$

> **``=value(Aire_cherchée);**

$$\begin{aligned} &= \left(\frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^3}{3} - 4 - \frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^2}{2} + \frac{5\sqrt{21}}{2} \right) u^2 + \left(-\frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^3}{3} \right. \\ &\quad \left. + \frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^3}{3} + \frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^2}{2} - \frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^2}{2} + 5\sqrt{21} \right) u^2 + \left(-\frac{25}{6} \right. \\ &\quad \left. - \frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^3}{3} + \frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^2}{2} + \frac{5\sqrt{21}}{2} \right) u^2 \end{aligned} \quad (15)$$

> **radnormal((15));**


$$= \frac{7(-7 + 6\sqrt{21})u^2}{6} \quad (16)$$

Exemple 3


Aire d'une région entre la courbe et l'axe des x d'une fonction non négative sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre la courbe $y = |x^2 - 6x + 8|$, l'axe des x et les droites d'équation $x = 1$ et $x = 5$.

```
> f:=x->abs(x^2-6*x+8);
Région:=plotcv([x,f(x),x=-0..6],x=1..5):
display(Région,view=[-1..7,0..4]);
```



```
> Rectangle1:=plotrectv([0,f(x)],x=1.5,delta=.5):
Rectangle2:=plotrectv([0,f(x)],x=3.5,delta=.5):
Rectangle3:=plotrectv([0,f(x)],x=4.7,delta=.5):
display([Rectangle1,Rectangle2,Rectangle3,Région],view=[-1..7,0..4]);
```



Calculons l'aire demandée. Les trois intégrandes que nous devrions poser sont identiques, alors

```
> Intégrande:=f(x)-0:
Aire3:=Int(Intégrande,x=4..5)*(u^2);
```

$$Aire3 := \left(\int_4^5 |x^2 - 6x + 8| dx \right) u^2 \quad (17)$$

```
> ``=value(Aire);
```

$$= 21 u^2 \quad (18)$$

Exemple 4


Aire d'une région entre la courbe et l'axe des x d'une fonction non négative sur un intervalle donné

Calcul de l'aire de la région comprise entre la courbe $y = (\sin^2)(x) \cos(x)$, l'axe des x et les droites


d'équation $x = -\frac{\pi}{2}$ et $x = \frac{\pi}{2}$

```
> f:=x->(sin^2)(x)*cos(x);
Région:=plotcv([x,f(x),x=-Pi..Pi],x=-Pi/2..Pi/2):
Région;
```

$$f := x \mapsto (\sin^2)(x) \cdot \cos(x)$$



```
> Rectangle1:=plotrectv([0,f(x)],x=-1,delta=.5):
Rectangle2:=plotrectv([0,f(x)],x=1.25,delta=.5):
display([Rectangle1,Rectangle2,Région]);
```



```
> Intégrande:=f(x)-0:
Aire:=Int(Intégrande,x=-Pi/2..Pi/2)*(u^2);
```

$$Aire := \left(\int_{-\frac{\pi}{2}}^{\frac{\pi}{2}} \sin(x)^2 \cos(x) \, dx \right) u^2 \quad (19)$$

```
> ``=value(Aire);
```

$$= \frac{2u^2}{3} \quad (20)$$

▼ Exemple 5

Aire d'une région fermée entre deux courbes sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre les courbes d'équation $y = \sin(x)$ et $y = \cos(x)$.

Déterminons les abscisses des points de rencontre entre les deux courbes.


```
> f:=x->sin(x):  
g:=x->cos(x):  
_EnvAllSolutions:=true:  
Sol:=solve(f(x)=g(x));
```

$$\text{Sol} := \frac{1}{4} \pi + \pi _ZI \quad (21)$$


Le domaine de la variable $_ZI$ est l'ensemble des entiers $Z = \{\dots, -2, -1, 0, 1, 2, 3, \dots\}$. Donc, sur la région d'intégration, avec $_ZI = 0, 1$ et 2 , les abscisses des points de rencontre sont $\frac{\pi}{4}$, $\frac{5\pi}{4}$ et $\frac{9\pi}{4}$.

Représentons les régions dont il faudra calculer l'aire.

```
> f:=x->sin(x);  
g:=x->cos(x);  
Région:=plot2cv([f(x),g(x)],x=0..5/2*Pi,x=Pi/4..9/4*Pi):  
Région;
```



```
> Rectangle1:=plotrectv([f(x),g(x)],x=2,delta=.5):  
Rectangle2:=plotrectv([f(x),g(x)],x=3.25,delta=.5):  
Rectangle3:=plotrectv([f(x),g(x)],x=5,delta=.5):  
Rectangle4:=plotrectv([f(x),g(x)],x=6,delta=.5):  
> display(seq(display(Rectangle | |k,Région),k=1..4),insequence=true);
```



Sélectionner le graphique précédent et exécuter l'animation une séquence à la fois.

Le calcul de l'aire cherchée peut être obtenu en additionnant deux intégrales définies. Créons les éléments différentiels d'aire pour chaque région et posons les intégrales définies.

> **Intégrande1:=f(x)-g(x):**

Aire1:=Int(Intégrande1,x=Pi/4..5/4*Pi)*(u^2);

$$Aire1 := \int_{\frac{\pi}{4}}^{\frac{5\pi}{4}} (\sin(x) - \cos(x)) dx u^2 \quad (22)$$

> **Intégrande2:=g(x)-f(x):**

Aire2:=Int(Intégrande2,x=5/4*Pi..9/4*Pi)*(u^2);

$$Aire2 := \int_{\frac{5\pi}{4}}^{\frac{9\pi}{4}} (-\sin(x) + \cos(x)) dx u^2 \quad (23)$$

Évaluons la somme des intégrales définies.

> **Aire:= Aire1 + Aire2;**

$$Aire := \int_{\frac{\pi}{4}}^{\frac{5\pi}{4}} (\sin(x) - \cos(x)) dx u^2 + \int_{\frac{5\pi}{4}}^{\frac{9\pi}{4}} (-\sin(x) + \cos(x)) dx u^2 \quad (24)$$

> **`` = value(Aire);**

$$= 4\sqrt{2} u^2 \quad (25)$$

Puisque les deux régions sont symétriques par rapport à l'axe $x = \frac{5\pi}{4}$, l'aire cherchée peut être obtenue directement en posant une seule intégrale définie.

> **Aire:=(2*Aire1);**

`` = value(Aire);

$$\begin{aligned} \text{Aire} &:= 2 \left(\int_{\frac{\pi}{4}}^{\frac{5\pi}{4}} (\sin(x) - \cos(x)) \, dx \right) u^2 \\ &= 4\sqrt{2} u^2 \end{aligned} \quad (26)$$

Exemple 6

Aire d'une région fermée entre deux courbes sur un intervalle à déterminer


Calcul de l'aire de la région comprise entre la courbe d'équation $y = x^2 + 5x - 2$ et la courbe d'équation $y = x - 4$.

Déterminons les abscisses des points de rencontre des deux courbes.


```
> f:=x->x^2+5*x-2:
g:=x->x-4:
Sol:=solve(f(x)=g(x));
evalf(Sol);
```

$$\begin{aligned} \text{Sol} &:= -2 + \sqrt{2}, -2 - \sqrt{2} \\ &-0.5857864380, -3.4142135620 \end{aligned} \quad (27)$$

```
> Région:=plot2cv([f(x),g(x),x=-4..1],x=min(Sol)..max(Sol)):
Région;
```



```
> Rectangle:=plotrectv([f(x),g(x)],x=-2,delta=.5):
display([Rectangle,Région],view=[-4..0,-9..-3]);
```




```
> Intégrande:=g(x)-f(x);
Aire:=Int(Intégrande,x=min(Sol)..max(Sol));
```

$$Aire := \int_{-2-\sqrt{2}}^{-2+\sqrt{2}} (-x^2 - 4x - 2) dx \quad (28)$$

```
> ``=value(Aire)*(u^2);
```

$$= \left(-\frac{(-2+\sqrt{2})^3}{3} + \frac{(-2-\sqrt{2})^3}{3} - 2(-2+\sqrt{2})^2 + 2(-2-\sqrt{2})^2 - 4\sqrt{2} \right) u^2 \quad (29)$$

```
> radnormal((29));
```

$$= \frac{8\sqrt{2} u^2}{3} \quad (30)$$


Exemple 7

Aire d'une région fermée entre deux courbes sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre les courbes f et g définies respectivement par $f(x) = \sin(x)$ et

$g(x) = \frac{1}{2}$ sur l'intervalle $\left[0, \frac{5\pi}{2}\right]$.

```
> f:=x->sin(x);
g:=x->1/2;
Région:=plot2cv([f(x),g(x),x=-Pi/2..5/2*Pi],x=0..5/2*Pi);
Région;
```



Trouvons les abscisses des points de rencontre des deux courbes.


```
> f:=x->sin(x); # _EnvAllSolutions est déjà initialisée à «true»
(Voir exemple 5)
g:=x->1/2;
Sol:=solve(f(x)=g(x),x);
f := x ↦ sin(x)
```

$$g := x \mapsto \frac{1}{2}$$

$$\text{Sol} := \frac{2}{3} \pi_{_B1} + \frac{1}{6} \pi + 2\pi_{_Z2} \quad (31)$$

Le domaine de la variable $_Z2$ est l'ensemble des entiers $\{\dots, -2, -1, 0, 1, 2, 3, \dots\}$ et le domaine de la variable $_B1$ est $\{0,1\}$. Donc, sur le domaine d'intégration, les abscisses des points de rencontre sont $\frac{\pi}{6}$, $\frac{5\pi}{6}$ et $\frac{13\pi}{6}$.

```
> Rectangle1:=plotrectv([f(x),g(x)],x=.25,delta=.5):
Rectangle2:=plotrectv([f(x),g(x)],x=2,delta=.5):
Rectangle3:=plotrectv([f(x),g(x)],x=5,delta=.5):
Rectangle4:=plotrectv([f(x),g(x)],x=7.5,delta=.5):
display([Rectangle1| | ($ 1..4),Région]);
```



Le calcul d'aire demandé doit donc se poser comme la somme de quatre intégrales définies.

```
> Intégrandel:=g(x)-f(x):
Aire1:=Int(Intégrandel,x=0..Pi/6)*(u^2):
Intégrande2:=f(x)-g(x):
Aire2:=Int(Intégrande2,x=Pi/6..5*Pi/6)*(u^2):
Intégrande3:=g(x)-f(x):
Aire3:=Int(Intégrande3,x=5*Pi/6..13/6*Pi)*(u^2):
Intégrande4:=f(x)-g(x):
Aire4:=Int(Intégrande4,x=13/6*Pi..5*Pi/2)*(u^2):
Aire:= Aire1 + Aire2 + Aire3 + Aire4;
```

$$\text{Aire} := \left(\int_0^{\frac{\pi}{6}} \left(\frac{1}{2} - \sin(x) \right) dx \right) u^2 + \left(\int_{\frac{\pi}{6}}^{\frac{5\pi}{6}} \left(\sin(x) - \frac{1}{2} \right) dx \right) u^2 + \left(\int_{\frac{5\pi}{6}}^{\frac{13\pi}{6}} \left(\frac{1}{2} - \sin(x) \right) dx \right) u^2 + \left(\int_{\frac{13\pi}{6}}^{\frac{5\pi}{2}} \left(\sin(x) - \frac{1}{2} \right) dx \right) u^2 \quad (32)$$

```
> ``=value(Aire);
```

$$= \left(-1 + \frac{\pi}{12} + \frac{\sqrt{3}}{2}\right) u^2 + \left(-\frac{\pi}{3} + \sqrt{3}\right) u^2 + \left(\frac{2\pi}{3} + \sqrt{3}\right) u^2 + \left(-\frac{\pi}{6} + \frac{\sqrt{3}}{2}\right) u^2 \quad (33)$$

```
> radnormal((33));
```

$$= -u^2 + \frac{u^2 \pi}{4} + 3u^2 \sqrt{3} \quad (34)$$

```
> sort(factor((34)));
```

$$= \frac{(-4 + \pi + 12\sqrt{3}) u^2}{4} \quad (35)$$

Exemple 8

Aire d'une région fermée entre deux courbes sur un intervalle à déterminer.

Calcul de l'aire de la région comprise entre la courbe $y_1 = x^4 + 1$ et la courbe $y_2 = 2x^2$.

Déterminons les abscisses des points de rencontre des deux courbes.


```
> f:=x->x^4+1;
g:=x->2*x^2;
Sol:=solve(f(x)=g(x));
```

$$f := x \mapsto x^4 + 1$$


$$g := x \mapsto 2 \cdot x^2$$

$$\text{Sol} := 1, 1, -1, -1 \quad (36)$$

```
> Région:=plot2cv([f(x),g(x),x=-1.5..1.5],x=min(Sol)..max(Sol)):
display(Région,view=[-1.5..1.5,0..4]);
```



```
> Rectangle:=plotrectv([f(x),g(x)],x=.5,delta=.25):
display([Rectangle,Région],view=[-1..1,0..2]);
```



Posons le calcul de l'aire cherchée

```
> Intégrande:=f(x)-g(x):
```

```
Aire:=Int(Intégrande,x=min(Sol)..max(Sol))*(u^2);
```

$$Aire := \left(\int_{-1}^1 (x^4 - 2x^2 + 1) dx \right) u^2 \quad (37)$$

Évaluons.

```
> ``=value(Aire);
```

$$= \frac{16u^2}{15} \quad (38)$$

Calcul d'aire d'une région plane selon un découpage horizontal

Exemple 1

Aire d'une région délimitée par une courbe et l'axe des y sur un intervalle donné


Calcul de l'aire de la région comprise entre la courbe d'équation $x = y^2 + 1$, l'axe des y , les droites d'équation $y = -2$ et $y = 3$.

Esquissons la région fermée entre les deux courbes.

```
> f:=y->y^2+1:
```


```
Région:=plotch([f(y),y,y=-3..4],y=-2..3):
```

```
Région;
```



Superposons un rectangle représentatif de l'élément différentiel d'aire horizontalement.

```
> Rectangle:=plotrecth([0,f(y)],y=2,delta=1):
display([Rectangle,Région]);
```



Pour intégrer par un découpage horizontal, il nous faut intégrer par rapport à y . Créons l'élément différentiel d'aire et posons l'intégrale définie.

```
> Intégrande:=f(y)-0:
Aire:=Int(Intégrande,y=-2..3)*(u^2);
```

$$\text{Aire} := \left(\int_{-2}^3 (y^2 + 1) \, dy \right) u^2 \quad (39)$$

```
> ``=value(Aire);
```

$$= \frac{50 u^2}{3} \quad (40)$$


Exemple 2

Aire d'une région délimitée par une courbe et l'axe des y sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre la courbe d'équation $x = 10 - 3y - y^2$, l'axe des y , $y = -7$ et $y = 3$.

Esquissons la région fermée entre les deux courbes.


```
> f:=y->10-3*y-y^2;
Région:=plot2ch([f(y),0,y=-8..4],y=-7..3):
Région;
```



```

> Rectangle1:=plotrecth([f(y),0],y=2.5,delta=1):
Rectangle2:=plotrecth([f(y),0],y=-3,delta=1):
Rectangle3:=plotrecth([f(y),0],y=-6,delta=1):
display([Rectangle1,Rectangle2,Rectangle3,Région]);

```



Obtenons l'ordonnée de chaque point de rencontre de la parabole avec l'axe des y.

```

> Sol:=solve(f(y)=0,y);

```

$$Sol := -5, 2$$

(41)

Calculons l'aire cherchée en additionnant l'aire de la surface de ces trois régions.

```

> Intégrale1:=0-f(y):

```

```

Aire1:=Int(Intégrale1,y=-7..min(Sol))*(u^2);

```

$$Aire1 := \left(\int_{-7}^{-5} (y^2 + 3y - 10) dy \right) u^2$$

(42)

```

> Intégrale2:=f(y)-0:

```

```

Aire2:=Int(Intégrale2,y=min(Sol)..max(Sol))*(u^2);

```

$$Aire2 := \left(\int_{-5}^2 (-y^2 - 3y + 10) dy \right) u^2$$

(43)

```

> Intégrale3:=0-f(y):

```

```

Aire3:=Int(Intégrale3,y=max(Sol)..3)*(u^2);

```

$$Aire3 := \left(\int_2^3 (y^2 + 3y - 10) dy \right) u^2$$

(44)

```

> Aire:=Aire1 + Aire2 + Aire3;

```

$$Aire := \left(\int_{-7}^{-5} (y^2 + 3y - 10) dy \right) u^2 + \left(\int_{-5}^2 (-y^2 - 3y + 10) dy \right) u^2 + \left(\int_2^3 (y^2 + 3y - 10) dy \right) u^2$$

(45)

```

> ``=value(Aire);

```

$$= \frac{233 u^2}{3}$$

(46)


Exemple 3

Aire d'une région délimitée par deux courbes sur un intervalle donné

Calcul de l'aire de la région comprise entre les courbes d'équations $x_1 = -y$ et $x_2 = 6y - y^2$ lorsque $y \in [1, 6]$.

Esquissons la région fermée entre les deux courbes.

```
> f:=y->-y;  
g:=y->6*y-y^2;  
Région:=plot2ch([f(y),g(y),y=0..7],y=1..6):  
Région;
```




Déterminons l'ordonnée de chaque point de rencontre entre les deux courbes.

```
> Sol:=solve(f(y)=g(y),y);
```

$Sol := 0, 7$

(47)

```
> Rectangle:=plotrecth([f(y),g(y)],y=5,delta=1):  
display([Rectangle,Région],view=[-10..10,-2..8]);
```



Posons l'intégrale définie.

```
> Intégrande:=g(y)-f(y):  
Aire:=Int(Intégrande,y=1..6)*(u^2);
```

$$Aire := \left(\int_1^6 (-y^2 + 7y) dy \right) u^2$$

(48)

Évaluons-la.

```
> ``=value(Aire);
```

$$= \frac{305 u^2}{6}$$

(49)

Exemple 4

Aire d'une région délimitée par deux courbes sur des intervalles à déterminer

Calcul de l'aire de la région comprise entre la courbe d'équation $x_1 = \frac{y^2}{2}$ et $x_2 = 4 + y$ lorsque $y \in [-4, 6]$.


Esquissons la région fermée entre les deux courbes et déterminons les ordonnées des points d'intersection.

```
> f:=y->y^2/2:
```

```
g:=y->4+y:
```

```
Région:=plot2ch([f(y),g(y),y=-4..6],y=-4..6):
```

```
Région;
```




```
> Rectangle1:=plotrecth([f(y),g(y)],y=5,delta=.75):
```

```
Rectangle2:=plotrecth([f(y),g(y)],y=2,delta=.75):
```

```
Rectangle3:=plotrecth([f(y),g(y)],y=-3,delta=.75):
```

```
display([Rectangle1,Rectangle2,Rectangle3,Région]);
```



Déterminons l'ordonnée de chaque point de rencontre entre les deux courbes.

```
> Sol:=solve(f(y)=g(y),y);
```

$$\text{Sol} := 4, -2$$

(50)

Calculons l'aire cherchée en additionnant l'aire de la surface des trois régions.

```
> Intégrandel:=f(y)-g(y):
Aire1:=Int(Intégrandel,y=-4..min(Sol))*(u^2);
```

$$Aire1 := \left(\int_{-4}^{-2} \left(-4 - y + \frac{1}{2} y^2 \right) dy \right) u^2 \quad (51)$$

```
> Intégrande2:=g(y)-f(y):
Aire2:=Int(Intégrande2,y=min(Sol)..max(Sol))*(u^2);
```

$$Aire2 := \left(\int_{-2}^4 \left(4 + y - \frac{1}{2} y^2 \right) dy \right) u^2 \quad (52)$$

```
> Intégrande3:=f(y)-g(y):
Aire3:=Int(Intégrande3,y=max(Sol)..6)*(u^2);
```

$$Aire3 := \left(\int_4^6 \left(-4 - y + \frac{1}{2} y^2 \right) dy \right) u^2 \quad (53)$$

```
> Aire:=Aire1 + Aire2 + Aire3;
```

$$Aire := \left(\int_{-4}^{-2} \left(-4 - y + \frac{1}{2} y^2 \right) dy \right) u^2 + \left(\int_{-2}^4 \left(4 + y - \frac{1}{2} y^2 \right) dy \right) u^2 + \left(\int_4^6 \left(-4 - y + \frac{1}{2} y^2 \right) dy \right) u^2 \quad (54)$$

```
> ``=value(Aire);
```


$$= \frac{98 u^2}{3} \quad (55)$$

Calcul d'aire d'une région plane selon les deux découpages

Calcul de l'aire de la région comprise entre la courbe d'équation $x_1 = y^2 - 3$ et la courbe d'équation $x_2 = y + 2$ selon un découpage horizontal et vertical.

Esquissons la région fermée entre les deux courbes.

```
> f:=y->y+2:
g:=y->y^2-3:
Région:=plot([[f(y),y,y=-3..3],[g(y),y,y=-3..3]],color=[navy,orange])
:
Région;
```



Effectuons d'abord le calcul avec un découpage horizontal en intégrant par rapport à y . Puisque x est déjà donnée explicitement en fonction de y dans les deux équations, il nous faut seulement déterminer les bornes d'intégration.

Déterminons l'ordonnée de chaque point de rencontre des deux courbes.

```
> Sol:=solve(f(y)=g(y),y);
```

$$Sol := \frac{1}{2} - \frac{\sqrt{21}}{2}, \frac{1}{2} + \frac{\sqrt{21}}{2} \quad (56)$$

Définissons les bornes d'intégration.

```
> Borne_inf:=min(Sol);
```


```
Borne_sup:=max(Sol);
```

$$Borne_inf := \frac{1}{2} - \frac{\sqrt{21}}{2}$$

$$Borne_sup := \frac{1}{2} + \frac{\sqrt{21}}{2} \quad (57)$$


```
> Région:=plot2ch([y+2,y^2-3,y=-3..3],y=Borne_inf..Borne_sup):
```

```
Région;
```



```
> Rectangle1:=plotrecth([f(y),g(y)],y=1,delta=1):
```

```
display([Rectangle1,Région]);
```



Créons l'élément différentiel d'aire et posons l'intégrale définie.

> Intégrande := f(y) - g(y) :

Aire := Int (Intégrande, y=Borne_inf..Borne_sup) * (u^2) ;

$$Aire := \left(\int_{\frac{1}{2} - \frac{\sqrt{21}}{2}}^{\frac{1}{2} + \frac{\sqrt{21}}{2}} (-y^2 + y + 5) dy \right) u^2 \quad (58)$$

> ``=value((58)) ;

$$= \left(-\frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^3}{3} + \frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^3}{3} + \frac{\left(\frac{1}{2} + \frac{\sqrt{21}}{2}\right)^2}{2} - \frac{\left(\frac{1}{2} - \frac{\sqrt{21}}{2}\right)^2}{2} + 5\sqrt{21} \right) u^2 \quad (59)$$

> radnormal((59)) ;

$$= \frac{7\sqrt{21} u^2}{2} \quad (60)$$

Calculons de nouveau l'aire de cette région en réalisant un découpage vertical.

Pour déterminer l'aire avec un découpage vertical, il nous faut intégrer par rapport à x . Il nous faut alors intégrer sur deux régions. Obtenons donc explicitement la variable y en termes de la variable x afin de faire la superposition des rectangles représentatifs des éléments différentiels d'aires.

> isolate(y=g(x), x) ;


$$x = \text{RootOf}(_Z^2 - y - 3) \quad (61)$$

Dans ce cas-ci, la macro-commande *solve* répondra mieux à nos besoins.


> solve(x=g(y), y) ;

$$\sqrt{x+3}, -\sqrt{x+3} \quad (62)$$

> Région ;



```
> Rectangle1:=plotrectv([sqrt(x+3),-sqrt(x+3)],x=-2,delta=1):
Rectangle2:=plotrectv([sqrt(x+3),x-2],x=1,delta=1):
display([Rectangle1,Rectangle2,Région]);
```



```
> Intégrandel:=sqrt(x+3)-(-sqrt(x+3)):
Aire1:=Int(Intégrandel,x=-3..f(Borne_inf))*(u^2);
```

$$Aire1 := \left(\int_{-3}^{\left(\frac{5}{2} - \frac{\sqrt{21}}{2}\right)} 2\sqrt{x+3} \, dx \right) u^2 \quad (63)$$

```
> Intégrande2:=sqrt(x+3)-(x-2):
Aire2:= Int(Intégrande2, x=f(Borne_inf)..f(Borne_sup))*(u^2);
```

$$Aire2 := \left(\int_{\left(\frac{5}{2} - \frac{\sqrt{21}}{2}\right)}^{\left(\frac{5}{2} + \frac{\sqrt{21}}{2}\right)} (\sqrt{x+3} - x + 2) \, dx \right) u^2 \quad (64)$$

```
> Aire:= Aire1 + Aire2;
```

$$Aire := \left(\int_{-3}^{\left(\frac{5}{2} - \frac{\sqrt{21}}{2}\right)} 2\sqrt{x+3} \, dx \right) u^2 + \left(\int_{\left(\frac{5}{2} - \frac{\sqrt{21}}{2}\right)}^{\left(\frac{5}{2} + \frac{\sqrt{21}}{2}\right)} (\sqrt{x+3} - x + 2) \, dx \right) u^2 \quad (65)$$

```
> ``=value((65));
```

$$= \left(4\sqrt{21} - \frac{32}{3} \right) u^2 + \left(\frac{32}{3} - \frac{\sqrt{21}}{2} \right) u^2 \quad (66)$$

```
| > radnormal((66));  
|  
|
```

$$= \frac{7\sqrt{21} u^2}{2}$$

(67)